2714 8
25 ... AT B

SAND77-0517
Unlimited Release

3151

The Environmental Data Processor of the Adaptive Intrusion Data System

MASTER

Michael S. Rogers

Sandia Laboratories

SF 2900-Q(7-73)

Issued by Sandia Laboratories, operated for the United States Energy Research & Development Administration by Sandia Corporation.

NOTICE

Printed in the United States of America

Available from
National Technical Information Service
U. S. Department of Commerce
5285 Port Royal Road
Springfield, VA 22161
Price: Printed Copy **$3.50;** Microfiche $3.00

SAND77-0517
Unlimited Release
Printed June 1977

THE ENVIRONMENTAL DATA PROCESSOR OF THE ADAPTIVE INTRUSION DATA SYSTEM

Michael S. Rogers
TM Development Division 9421
Sandia Laboratories
Albuquerque, NM 87115

ABSTRACT

A data acquisition system oriented specifically toward collection and processing of various meteorological and environmental parameters has been designed around a National Semiconductor IMP-16 microprocessor. This system, called the Environmental Data Processor (EDP), was developed specifically for use with the Adaptive Intrusion Data System (AIDS) in a perimeter intrusion alarm evaluation, although its design is sufficiently general to permit use elsewhere. This report describes in general detail the design of the EDP and its interaction with other AIDS components.

CONTENTS

THE ENVIRONMENTAL DATA PROCESSOR OF THE ADAPTIVE INTRUSION DATA SYSTEM

Introduction

The advent of the microprocessor has made practical many concepts heretofore deemed unfeasible because of the economic and/or engineering investment required for a suitable implementation. One such application is to small, moderately powerful, dedicated acquisition systems for which minicomputers represent overkill and for which random logic is too complex and unwieldy.

The microprocessor-based system to be described was developed specifically for a perimeter intrusion alarm evaluation conducted by Sandia at Rocky Flats, Colorado, and at the Sandia Alarm Development Laboratory (SADL) in Area III, Building 6600, Sandia Laboratories, Albuquerque, New Mexico. However, its design is sufficiently general to permit utilization in other applications.

The system was designed around a National Semiconductor IMP-16 microprocessor for several reasons. At the time of initial system definition, the IMP-16 was one of the few proven microprocessors on the market. Also, the IMP's instruction repertoire includes such single-instruction commands as multiply and divide, which has proved very useful in implementing some of the mathematical algorithms required for processing much of the data. Also, the 16-bit word size in the IMP was attractive because most other system components utilized a 16-bit word structure, and all interconnecting data buses were 16-line. The IMP-16 has proved to be a powerful system tool in this application and has functioned reliably. However, some of the newer generation microprocessors may possibly be better choices in that they require much less overhead logic and have whole families of specialized support circuits that greatly simplify both hardware and software. Nonetheless, the IMP-based unit has demonstrated the feasibility of microprocessor-based acquisition systems. Flexibility, adaptability, computational power, and susceptibility to miniaturization are all characteristics of a microprocessor-based system which give it a distinct and commanding advantage over random logic systems.

Network Description

The overall network of which the Environmental Data Processor (EDP) is a part is shown in Figure 1. This network is known as the Adaptive Intrusion Data System (AIDS), and most of the functional blocks shown in Figure 1 are described in other Sandia reports. Table I lists the meteorological parameters currently being collected by the EDP at Rocky Flats, Colorado. As shown in Figure 1, most of the various sensors are located on a 10-foot tower, except for the potential gradient sensor, which is placed on the ground, and the tipping-bucket rain gage, which is located on the roof of the instrumentation trailer. Also, at Rocky Flats, the two remote wind speed and wind direction sensors are mounted on two additional towers on which cameras are also mounted. All of the environmental sensors and sensor processors are off-the-shelf commercial units.

The raw sensor outputs are converted to easily digitized analog levels by the sensor processors. The sensor processors also provide regulated power to the sensors when such power is necessary for operation. The analog outputs of the sensor processors in turn feed the EDP, where each is sampled 10 times per second, digitized, scaled, and further processed as appropriate to the particular data parameter. The EDP converts all input data to proper engineering units and formats the data for display and subsequent storage on magnetic tape. The processed metro data are transferred from the EDP's memory on command to the Memory Controlled Processor (MCP), which interposes the environmental channels with other data. The composite data stream is then directed into a mass memory where, depending upon various criteria, it is either discarded or transferred to magnetic tape for permanent storage.



Figure 1. Overall Adaptive Intrusion Data System Diagram

TABLE I

AIDS Metro Data Rocky Flats

Channel	Data, Location	Units	Range
0	Temperature, Trailer	°F	-30 to +130
1	Wind Speed, Trailer	Miles per hour	0 to 100
2	Wind Direction, Trailer	Degrees azimuth	0 to 360
3	Relative Humidity, Trailer	%	0 to 100
4	Precipitation Rate, Trailer	Inches per hour	0 to 9.99
5	Potential Gradient, Trailer	(Volt/meter) ÷10	-999 to +999
6	Precipitation Accumulator, Trailer	Hundredths of an inch	0 to 255
7	Wind Speed, Remote, Corner Sectors 4 & 5	Miles per hour	0 to 100
8	Wind Direction, Remote, Corner Sectors 4 & 5	Degrees azimuth	0 to 360
9	Wind Speed, Remote, South End, Sector 3	Miles per hour	0 to 100
10	Wind Direction, Remote, South End, Sector 3	Degrees azimuth	0 to 360

All data are right justified, straight binary coded. Bit 11 is the sign bit for all channels. "1" indicates that data are positive (+), and "0" indicates that data are negative (-). For location purposes, Bit 0 = LSB is rightmost bit and Bit 11 = MSB is leftmost bit.

Environmental Data Processor Description

Figure 2 is a detailed block diagram of the EDP. The outputs of the commercially supplied sensor processors are fed differentially to the EDP; i.e., both signal high (+) and signal low (-) are transmitted via a twisted-pair cable. The front end of the EDP consists of IC instrumentation amplifiers which buffer and differentially amplify the sensor processor outputs. This scheme helps minimize common-mode signal errors resulting from spurious interference on the signal lines.

As evident from Figure 2, the analog inputs from the metro processors are routed through a multipole relay into instrumentation amplifiers. The analog output of a calibration circuit is also fed to contacts on this relay, and the amplifier inputs can be switched between data and calibration upon command from the System Controller (a Nova 2 minicomputer). Ordinarily, of course, the amplifier inputs are connected to a data source, but periodically a calibration sequence is run. The information obtained in a calibration run can provide valuable assistance in interpreting recorded data should the system malfunction noncatastrophically. For instance, if an amplifier drifts or its gain deviates, or if any system nonlinearities become pronounced, examination of the known calibration values can provide a basis for adjustment of the recorded data to yield acceptable accuracy, at least until the malfunction can be isolated and corrected in the field.


Clock
Addr. Strb.
Data Strb.
Bus Grant
Bus Req.
Write Mem.
Cmd.
Interface Control Logic
Start Conv.
End Conv.
12-Bit A/D
To Metro Processor 1
Sig.
To Metro Processor 16
16 Channel Mux
Mux Addr Ctr
16-Bit, 3-State Bus Driver (Address)
Enable
16-Bit, 3-State Bus Driver (Data)
4-Bit Misc Info
9-Bit D/A Conv.
Data
Cal Signal
To System Controller (Nova 2 Mini)
Cal Cmd.
Start
Reply
Stop
Control Logic
Relay Driver
To Memory Controlled Processor
Data
Load
Init.
Data Strb.
Read Mem.
12-Bit Bus Driver (Data)
Clk
Preset
4-Bit, Binary Counter
16-Bit, 3-State Bus Driver (Address)
Display Strobes
Gate
4 to 16 Decoder
Front Panel

1



Figure 2. EDP Block Diagram



Digital Display
4-Bit
Data
Strobe
Latch
BCD to 7-Seg. De-coder
Timing & Control Bus
Decoded Write Mem. Cmd.
Write
Periph. Cmd.
Data Strb.
Addr. Strb.
To Other Displays
3-Digit Decimal Display
Receiver Enable
Driver Enable
16-Bit Bidirectional Transceiver
Read/Write Memory
PROM Program Storage
Clock
CPU
Init.
Run
Restart
16-Bit Address & Data Bus

3

The outputs of the instrumentation amplifiers feed into a 16-channel multiplexer where they are concentrated by time-division-multiplex onto a single line. This line in turn feeds a 12-bit, analog-to-digital converter. The 12-bit output of the converter is tied to a tristate buffer, where it is combined with four bits of miscellaneous information (sign bit, digital filter control, timed pulse, etc.). The end-of-conversion signal from the A/D converter initiates a bus request to the IMP-16 CPU, which responds with a bus grant signal when the 16-bit address/data bus becomes free and no higher-priority bus requests are active (the IMP utilizes a four-level priority for bus control). When the bus grant is received, it triggers a sequence of commands and signals which accomplish the data transfer. First, a write memory command is generated and passed to the CPU, which informs the IMP that data will be written into its read/write memory during the present bus cycle. Next, the interface control logic outputs an enable for the 16-bit, tristate address buffer. (See Figure 2; the IMP has a common 16-bit, tristate bus for both address and data, which are time-multiplexed (address first, then data) onto the bus. The tristate structure of the bus allows many peripherals to be connected directly to the bus through separate buffers.) The address buffer is fed by the four-bit address counter from the multiplexer, which contains the location of the current channel (input) connected to the A/D converter. This address is placed on the bus and, when combined with appropriate signals from the CPU, "opens" the memory location where the data will be stored (raw, unprocessed data from the 16 input channels are placed in memory locations 0 through 15). Immediately following the address strobe, a data buffer enable is generated which strobes the 16-bit data (12 A/D bits, 4 miscellaneous bits) onto the bus and consequently into the "opened" memory location. The entire transaction described occurs within 1.04 μs, the length of a bus cycle. The bus cycle extends from removal of the bus grant until termination of the write memory command. The basic 1.04-μs length can be extended if necessary to facilitate use of slow access memories. Understanding of the data transfer between A/D and memory may be enhanced by referring to the fundamental timing diagram shown in Figure 3.



Figure 3. Interface Timing Diagram

The technique of writing data directly into central memory via a Direct Memory Access (DMA) scheme, independent of software control, instead of interrupting the CPU and inserting the raw data directly into the respective software routine, was chosen because of the ease of implementation, speed, and flexibility which it afforded. The insertion of digitized, unprocessed inputs directly into memory allows them to be retrieved by the individual data routines during the normal sequence of execution without the need to incorporate interrupt service routines for each data input.

As each data parameter is processed by the IMP, it is again stored in memory as well as output to a display. The display panel consists of eight 3-digit LED displays. Six of the displays are dedicated, but the remaining two can accommodate any of the 16 possible data channels. The desired parameter for display is selected via a 16-position thumbwheel switch mounted on the front panel. To minimize cabling between the front panel and the logic card backplane, the displays are multiplexed, with common data lines and a separate strobe line tied to each three-digit display.

The display hardware, as well as the logic contained inside the LED display digit, is shown in block form in Figure 2. Data output to the displays is accomplished by use of an output instruction in the IMP software. The processed data are sent directly to the displays from a register in the IMP CPU and are not transferred to the displays from memory (although the data are also stored in memory). The software I/O instruction automatically generates all necessary timing and control signals, which the selected peripheral (the group of displays) decodes and uses to strobe data into the proper display at the correct time.

Since most of the meteorological data vary slowly, they are sent to the displays immediately after being processed by the respective software routines. However, to minimize display flicker caused by rapidly changing wind speed and direction, the latter parameters are smoothed through a digital filter subroutine before being sent to the displays. The filter subroutine, as well as all other software routines, is explained further in the appendix, which contains a complete listing of the Rocky Flats Environmental Data Processor control program.

The processed environmental data must also be made available for permanent storage on magnetic tape, since automatic recording is essential for the large amounts of data handled by the system of which the EDP is a part. However, the EDP itself does not have a direct tape interface, but passes its data to a Memory Controlled Processor (MCP; see Figure 1). The MCP in turn combines the environmental data with all other collected data and formats the composite for tape storage.

The EDP-MCP interface is shown in Figure 2. The processed environmental data to be transferred to the MCP are temporarily stored in read/write memory in the EDP. When the MCP initiates a data transfer, it first issues an initialization pulse to the EDP. This pulse presets a counter which is connected to the IMP address/data bus via a tristate bus driver. After setup, the MCP then issues a load pulse every 8 μs until all data words are transferred. The load pulse steps the address counter and also triggers a bus request to the IMP. When a bus grant is received from the microprocessor, the address of the desired word is placed onto the IMP bus. When the address portion of the bus cycle is terminated (see Figure 3), the data appear on the bus. After the data have settled, the EDP generates a strobe pulse for the MCP to use to latch the word into its data buffer. The MCP keeps track of the number of active environmental channels and, when all data words have been transferred, it terminates the transaction by inhibiting further load pulses.

Packaging

Figure 4 shows the data display and logic chassis containing most of the Sandia-designed circuitry utilized in the EDP. Figure 5 shows the full 4-bay rack which houses all of the AIDS system.



Figure 4. EDP Logic and Display Chassis



Figure 5. AIDS System Hardware

APPENDIX

This appendix contains a complete listing (with comments) of the control software for the Environmental Data Processor (EDP) located at Rocky Flats, Colorado. Since it was initially anticipated that the microprocessor in the EDP might be used to perform other functions as the overall system was developed, more memory (both read-only memory, ROM, and read/write memory) than was necessary for the EDP was included to accommodate subsequent expansion of the system. A total of 8,000 active memory locations are available for software and data storage in the Rocky Flats EDP, including 4K of ROM for program storage and 4K of read/write memory for scratchpad, temporary storage, and data storage.

```
TYPE ADS2
[15:42:32]
;        ADAPTIVE INTRUSION DATA SYSTEM (AIDS) ENVIRONMENTAL DATA
;        PROCESSOR SYSTEM PROGRAM TO RUN ON NATIONAL SEMICONDUCTOR
;        IMP 16L MICROPROCESSOR. THIS PROGRAM PROVIDES SOFTWARE CONTROL
;        FOR THE ENVIRONMENTAL PROCESSOR LOCATED AT ROCKY FLATS, COLO.
;
;        PROGRAM IS STORED IN READ ONLY MEMORY (ROM) ON A 4K
;        ROM MEMORY BOARD RESIDENT IN THE 16L CHASSIS. THE
;        PROGRAM BEGINS AT ROM MEMORY LOCATION 61440 (F000 HEX). THE
;        4096 RAM MEMORY IS USED FOR DATA STORAGE AND SCRATCHPAD.
;
;        INPUT DATA TO THE MICROPROCESSOR IS FROM A 12-BIT ANALOG TO
;        DIGITAL CONVERTER.  WHEN THE FULL 12 BITS IS NOT NEEDED FOR
;        DATA REPRESENTATION, THE DATA IS TRUNCATED IN THE PROCESSOR
;        BY RIGHT SHIFTING.  INPUT SAMPLING RATE FOR THE A/D IS 10/SEC
;        PER CHANNEL.
;
;
;        CHANNEL 0 TEMPERATURE ROUTINE FOLLOWS.  AS TEMP GOES FROM
;        -30 TO +130 DEGREES F, INPUT TO THE MICROPROCESSOR GOES FROM
;        0 TO 160.  SOFTWARE DETECTS THE CROSSOVER AND ADJUSTS THE
;        DATA ACCORDINGLY.
;
         .EXTD                     ;SUPPRESS ERROR MESSAGES IN ASSEM-
;                                   BLY RESULTING FROM USE OF EXTENDED
;                                   INSTRUCTION SET.
P:       .=.+61440                 ;BIAS ASSEMBLER TO 1ST MEMORY LOCATION.
START:   LD       0,0              ;FETCH TEMP DATA.
         SHR      0,4              ;TRUNCATE TO 8 BITS.
         ST       0,64
         LI       0,30             ;SET AC0=30.
         SKG      0,64             ;SKIP NEXT INST IF TEMP<0.
         JMP      POS
         SUB      0,64             ;SUBTRACT 30-DATA.
         JMP      OUT
POS:     LD       0,64
         SUB      0,D30            ;SUBTRACT DATA-30.
         SETBIT   11               ;BIT 11 OF AC0 = 1 INDICATES DATA
;                                   IS POSITIVE.
OUT:     ST       0,16             ;STORE ADJUSTED DATA.
         CLRBIT   11
         JSR      @BNBCD           ;GO TO BINARY TO BCD CONVERSION ROUTINE.
         LI       1,0              ;SET AC1 TO ADDRESS TEMP DISPLAY.
         LD       2,64
         SKG      2,D30            ;SKIP NEXT INST IF TEMP>0.
         ADD      0,HD00           ;SET MINUS IN DIGIT 3.
         JSR      @OUTPT
;
;
;        CHANNEL 1,7, AND 9 WIND SPEED AND CHANNEL 2,8, AND 10 WIND DIR-
;        ECTION ROUTINES FOLLOW. DATA IN THESE CHANNELS IS DIGITALLY
;        FILTERED BEFORE OUTPUTING TO THE DISPLAYS. THE DIGITAL FIL-
;        TER CALL IS INITIATED BY A CONTROL BIT (BIT 14) IN THE WIND
;        SPEED CHANNEL 1 INPUT. THIS CONTROL BIT IS IN TURN DERIVED FROM
;        A 1-PULSE/SEC SIGNAL FROM THE TIME CODE TRANSLATER. THE CONTROL
;        BIT IN EFFECT DETERMINES THE SAMPLE RATE OF THE FILTER AND SETS
;        THIS RATE AT 1-SAMPLE/SEC VS 10/SEC FOR THE CHANNEL INPUTS. DATA
;        STORED IN MEMORY FOR WRITING ON MAGNETIC TAPE IS NOT FILTERED.
```

```
        LD      0,1             ;FETCH W.S. DATA (CHANNEL 1).
        SKBIT   14              ;SKIP NEXT INST IF CONTROL BIT = 1,I.E.,
;                                COMMAND TO SAMPLE IS ACTIVE.
        JMP     .+5             ;SKIP NEXT 4 INST IF BIT 14 = 0.
        PFLG    0               ;RESET CONTROL F-F.
        SETST   10              ;SET STATUS FLAG 10.
        CLRBIT  14              ;REMOVE CONTROL BIT.
        ST      0,1             ;REWRITE MEMORY AFTER CLEARING CONTROL
;                                BIT SO FILTER SUBROUTINE WILL NOT BE
;                                CALLED ERRONEOUSLY.
        SHR     0,5             ;TRUNCATE W.S. TO 7 BITS.
        ST      0,65
        SETBIT  11              ;DATA POSITIVE.
        ST      0,17            ;STORE INSTANTANEOUS W.S. (CHANNEL 1).
        LD      0,7             ;FETCH W.S. (CHAN 7).
        SHR     0,5
        ST      0,70
        SETBIT  11              ;DATA POSITIVE.
        ST      0,23            ;STORE CHANNEL 7 W.S.
        LD      0,9             ;FETCH W.S. (CHAN. 9).
        SHR     0,5
        ST      0,75
        SETBIT  11
        ST      0,25            ;STORE W.S. (CHAN 9).
        LD      0,2             ;FETCH W.D. (CHAN 2).
        SHR     0,3             ;TRUNCATE TO 9 BITS.
        ST      0,80
        SETBIT  11
        ST      0,18            ;STORE INSTANTANEOUS W.D. (CHAN 2).
        LD      0,8             ;FETCH W.D. (CHAN 8).
        SHR     0,3
        ST      0,85
        SETBIT  11
        ST      0,24            ;STORE W.D. (CHAN 8).
        LD      0,10            ;FETCH W.D. (CHAN 10).
        SHR     0,3
        ST      0,90
        SETBIT  11
        ST      0,26            ;STORE W.D. (CHAN 10).
        SKSTF   10              ;SKIP NEXT INST IF FILTER IS TO
;                                BE CALLED.
        JMP     HUM             ;GO TO RELATIVE HUMIDITY ROUTINE.
        CLRST   10              ;DISABLE FURTHER FILTER CALLS.
        LI      2,65            ;AC2=65. THIS PUTS A BASE ADDRESS IN AC2
;                                FOR USE IN INDEXED ADDRESSING IN THE
;                                FILTER SUBROUTINE.
        JSR     FILT            ;GO TO FILTER WITH W.S.
        JSR     BINBCD          ;GO TO BINARY TO BCD CONV SUBROUTINE
;                                UPON RETURN FROM FILTER.
        LI      1,1             ;SET AC1 TO ADDRESS W.S. DISPLAY.
        JSR     OUTPUT
        LI      2,70            ;SET AC2 FOR INDEXED ADDRESSING FOR
;                                W.D. IN FILTER.
        JSR     FILT
        JSR     BINBCD
        LI      1,7             ;SET AC1 TO ADDRESS W.S. (CHAN 7)
;                                DISPLAY.
        JSR     OUTPUT
        LI      2,75
        JSR     FILT
```

```
        JSR     BINBCD
        LI      1,9             ;SET AC1 TO ADDRESS CHAN 9 W.S. DISPLAY.
        JSR     OUTPUT
        LI      2,80
        JSR     FOLD
        JSR     FILT
        JSR     TEST
        JSR     BINBCD
        LI      1,2             ;SET AC1 TO ADDRESS W.D. DISPLAY
;                                (CHANNEL, 2.
        JSR     OUTPUT
        LI      2,85
        JSR     FOLD
        JSR     FILT
        JSR     TEST
        JSR     BINBCD
        LI      1,8             ;SET AC1 TO ADDRESS CHAN 8 W.D.
        JSR     OUTPUT
        LI      2,90
        JSR     FOLD
        JSR     FILT
        JSR     TEST
        JSR     BINBCD
        LI      1,10            ;SET AC1 TO ADDRESS CHAN 10 W.D.
        JSR     OUTPUT
;
;
;       CHANNEL 3 RELATIVE HUMIDITY ROUTINE FOLLOWS. R.H. SENSOR BECOMES
;       NONLINEAR WHEN TEMPERATURE FALLS BELOW 30 DEGREES F. THIS ROU-
;       TINE APPROXIMATELY LINEARIZES THE R.H. READING BELOW 30 F.
;
HUM:    LD      0,0             ;FETCH TEMP.
        SHR     0,4
        LD      1,3             ;FETCH R.H.
        SHR     1,5
        ST      1,150
        SKG     0,D60           ;SKIP NEXT INST IF TEMP>30.
        JMP     NEXT
        JMP     STORE
NEXT:   SKG     0,D40           ;SKIP NEXT INST IF TEMP>10.
        JMP     NEXT1
        ADD     1,D4            ;ADD R.H. + 4.
        JMP     STORE
NEXT1:  SKG     0,D20           ;SKIP IF TEMP>-10.
        JMP     NEXT2
        MPY     D14             ;THIS INST AND FOLLOWING 3 INST PER-
        ADD     1,D418          ;FORM LINEARIZATION BY FITTING R.H. TO
        DIV     D100            ;EQUATION: RH' = RH + .14(RH) + 4.18.
        ADD     1,150
        JMP     STORE
NEXT2:  MPY     D26             ;THIS AND NEXT 3 INST PERFORM LINEAR-
        ADD     1,D786          ;IZATION BY FITTING R.H. TO EQUATION
        DIV     D100            ;RH' = RH + .26(RH) + 7.86.
        ADD     1,150
STORE:  RCPY    1,0
        SETBIT  11              ;DATA POSITIVE.
        ST      0,19            ;STORE PROCESSED R.H.
        CLRBIT  11
```

```
          JSR      BINBCD
          LI       1,3                  ;SET AC1 TO ADDRESS R.H. DISPLAY.
          JSR      OUTPUT
          JMP      RAIN
;
;
;         CONSTANT VALUES USED IN TEMPERATURE, WIND SPEED, WIND DIRECTION,
;         AND RELATIVE HUMIDITY ROUTINES AND BINARY TO BCD CONVERSION
;         SUBROUTINE ARE LISTED BELOW. CONSTANTS WHOSE LABEL BEGINS WITH
;         A "D" ARE DECIMAL VALUES, WHILE THOSE BEGINNING WITH AN "H" ARE
;         HEXADECIMAL VALUES.
;
D30:      .WORD    30
HD00:     .WORD    0D00
BNBCD:    .WORD    61584
OUTPT:    .WORD    61604
D60:      .WORD    60
D40:      .WORD    40
D4:       .WORD    4
D14:      .WORD    14
D418:     .WORD    418
D26:      .WORD    26
D786:     .WORD    786
D100:     .WORD    100
H3000:    .WORD    03000
H4FFF:    .WORD    04FFF
D20:      .WORD    20
;
;
;         BINARY TO BCD CONVERSION SUBROUTINE FOLLOWS. THIS SUBROUTINE
;         ACCEPTS A BINARY INPUT IN AC0 AND RETURNS UP TO 4 DIGITS OF
;         BCD IN AC0. IT UTILIZES THE ADD 3, SHIFT LEFT TECHNIQUE FOR
;         CONVERSION.
;
BINBCD:   LI       1,0                  ;INITIALIZE REGISTERS.
          RXCH     0,1
          LI       2,16                 ;SET SHIFT COUNT.
L1:       JSR      UPDEC                ;CALL ROUTINE TO CHECK BCD DECADES.
          SFLG     2                    ;INCLUDE LINK BIT IN SHIFT AND ROTATE.
          SHL      1,1                  ;SHIFT FROM BINARY REGISTER TO
          ROL      0,1                  ;BCD REGISTER.
          AISZ     2,-1                 ;DECREMENT SHIFT COUNT.
          JMP      L1                   ;CONTINUE.
          RTS      0                    ;RETURN FROM SUBROUTINE TO MAIN PROGRAM.
UPDEC:    BOC      1,RET                ;SKIP ROUTINE TO CHECK BCD DECADES IF
;                                        DECADE = 0.
          PFLG     2                    ;EXCLUDE LINK BIT IN ROTATES.
          LI       3,4                  ;SET BCD DECADE COUNT.
L2:       SKG      0,H4FFF              ;SKIP NEXT INST IF DECADE VALUE > 4.
          BOC      2,.+2
          ADD      0,H3000              ;ADD 3 TO DECADE.
          ROL      0,4                  ;POSITION NEXT DECADE.
          AISZ     3,-1                 ;DECREMENT COUNT AND SKIP IF 0.
          JMP      L2
RET:      RTS      0
;
;
;         DISPLAY OUTPUT SUBROUTINE FOLLOWS. THIS ROUTINE INVOKES THE
;         NECESSARY INSTRUCTIONS TO OUTPUT DATA FROM THE MICROPROCESSOR
;         TO THE DISPLAYS.
```

```
OUTPUT: SHL     1,12            ;POSITION DISPLAY ADDRESS.
        RADD    1,0             ;PUT DISPLAY ADDRESS IN AC0.
        LI      3,0             ;INITIALIZE AC3 FOR I/O.
        ROUT    8               ;OUTPUT DATA TO DISPLAY.
        RTS     0               ;RETURN.
;
;
;       SCALE FOLDING SUBROUTINE FOR WIND DIRECTION FOLLOWS. THIS ROU-
;       TINE ESSENTIALLY FOLDS THE W.D. SCALE SUCH THAT OUTPUT GOES FROM
;       0-180-0 DEGREES AS INPUT GOES FROM 0-360. FOLDING THE SCALE
;       PERMITS FILTERING AND ELIMINATION OF THE 0,360 CROSSOVER AMBI-
;       GUITY. THIS SUBROUTINE IS INVOKED ONLY IF W.D. > 180 DEGREES.
;
FOLD:   LD      1,(2)           ;FETCH W.D. DATA.
        SKG     1,D180          ;SKIP NEXT INST IF W.D.>180 DEGREES.
        RTS     0               ;EXIT.
        LD      1,D360
        SKG     1,(2)           ;SKIP IF W.D.<360.
        ST      1,(2)           ;SET W.D.=360. THIS ELIMINATES SUBSE-
;                                QUENT PROBLEMS WHICH ARISE IF BIT
;                                FLICKER MAKES W.D.>360.
        SUB     1,(2)           ;SUBTRACT 360-W.D.
        ST      1,(2)           ;STORE FOLDED DATA IN BASE LOCATION FOR
;                                WIND DIRECTION.
        SETST   11              ;SET FLAG 11 IF IN SECOND PORTION OF
;                                FOLDED SCALE, I.E., W.D.>180.
        RTS     0               ;RETURN TO MAIN ROUTINE.
;
;
;       TEST SUBROUTINE FOR WIND DIRECTION FOLLOWS. THIS ROUTINE TESTS
;       THE W.D. DATA AFTER FILTERING TO DETERMINE IF THE SCALE HAS
;       BEEN FOLDED, I.E., IF W.D.>180 DEGREES. IF SO, IT UNFOLDS THE
;       SCALE SO OUTPUT DATA WILL AGAIN RANGE FROM 0-360 DEGREES.
;
TEST:   SKSTF   11              ;SKIP NEXT INST IF FLAG 11 IS SET,
;                                I.E., IF W.D.>180.
        RTS     0               ;EXIT
        LD      0,D360
        SUB     0,1(2)          ;SUBTRACT 360-W.D. AFTER FILTERING.
        CLRST   11              ;CLEAR 180-0 FLAG FOR NEXT ITERATION.
        RTS     0               ;RETURN TO MAIN PROGRAM.
;
;
;       CONSTANTS FOR THE SCALE FOLDING ("FOLD") AND TEST SUBROUTINES
;       ARE LISTED BELOW. THE LABELS FOR THESE CONSTANTS ARE PREFIXED
;       WITH A "D" TO INDICATE THAT THEY ARE DECIMAL VALUES.
;
D180:   .WORD   180
D360:   .WORD   360
;
;
;       DIGITAL FILTER SUBROUTINE FOLLOWS. THIS FILTER IS A 2-POLE,
;       LOWPASS BUTTERWORTH REALIZATION WITH A CUTOFF FREQUENCY FC=.1
;       HZ. THIS IS A RECURSIVE FILTER AND IS DESCRIBED BY THE FOLLOWING
;       ALGORITHM:
;
;       G(M) = .0675(F(M)+2F(M-1)+F(M-2))-(.4128G(M-2)-1.143G(M-1))
```

```
;       WHERE,
;
;       G(M) = PRESENT VALUE OF OUTPUT AT TIME MT.
;       G(M-1) = 1ST PAST VALUE OF OUTPUT AT TIME (M-1)T.
;        G(M-2) = 2ND PAST VALUE OF OUTPUT AT TIME (M-2)T.
;       F(M) = PRESENT VALUE OF INPUT AT TIME MT.
;       F(M-1) = 1ST PAST VALUE OF INPUT AT TIME (M-1)T.
;       F(M-2) = 2ND PAST VALUE OF INPUT AT TIME (M-2)T.
;       T = SAMPLING INTERVAL = 1 SEC.
;       M = NUMBER OF SAMPLES TAKEN (ZERO TO INFINITY).
;
;       THE FILTER SUBROUTINE RETURNS THE PRESENT OUTPUT VALUE, I.E.,
;       G(M), IN ACØ.
;       SINCE THE IMP 16 DOES NOT HAVE FLOATING POINT MATH CAPABILITY,
;       THE FILTER COEFFICIENTS ARE MULTIPLIED BY 10,ØØØ AND THEN G(M)
;       IS DIVIDED BY 10,ØØØ AFTER THE ALGORITHM IS COMPLETED.
;       SINCE SUBROUTINE FILT CAN BE CALLED BY ANY PRIME ROUTINE, AND
;       SINCE SEVERAL PAST VALUES MUST BE STORED FOR EACH CALLING
;       ROUTINE, IT IS NECESSARY TO USE AN INDEXED ADDRESSING SCHEME
;       SO THAT DATA FOR THE VARIOUS CALLING ROUTINES CAN BE RETRIEVED
;       AND STORED IN SEPARATE MEMORY LOCATIONS. THIS IS ACCOMPLISHED
;       BY USING AC2 AS A BASE REGISTER AND HAVING EACH CALLING ROUTINE
;       LOAD A DIFFERENT BASE ADDRESS INTO AC2. SUBROUTINE FILT CAN
;       THEN APPROPRIATE MEMORY BY INDEXING THIS BASE ADDRESS.
;
FILT:   LD      1,1(2)          ;LOAD AC1 WITH G(M-1).
        MPY     B               ;MULTIPLY B X G(M-1).
        ST      Ø,148           ;STORE HI ORDER RESULT IN LOCATION 148.
        ST      1,149           ;STORE LO ORDER RESULT IN 149.
        LD      1,2(2)          ;LOAD AC1 WITH G(M-2).
        MPY     C               ;MULTIPLY C X G(M-2).
        DSUB    148             ;DOUBLE PRECISION SUBTRACT C X G(M-2) -
;                                B X G(M-1).
        ST      Ø,148
        ST      1,149
        LD      1,3(2)          ;LOAD AC1 WITH F(M-1).
        SHL     1,1             ;MULTIPLY 2 X F(M-1).
        ADD     1,(2)           ;ADD 2 X F(M-1) + F(M).
        ADD     1,4(2)          ;ADD (2 X F(M-1) + F(M)) + F(M-2).
        MPY     A               ;MULTIPLY ABOVE BY COEFFICIENT A.
        DSUB    148             ;DOUBLE PRECISION SUBTRACT
;                                A(F(M)+2F(M-1)+F(M-2)) -
;                                (C X G(M-2)-B X G(M-1)).
        DIV     D1ØØØØ          ;DIVIDE RESULT BY 1ØØØØ.
        SKG     Ø,D5ØØØ         ;SKIP NEXT INST IF REMAINDER > 5ØØØ.
        JMP     SAVE
        AISZ    1,1             ;ADD 1 TO RESULT (ROUNDOFF).
;       FOLLOWING STATEMENTS SAVE PAST VALUES.
SAVE:   LD      3,3(2)          ;LOAD AC3 WITH F(M-1).
        ST      3,4(2)          ;REPLACE F(M-2) WITH F(M-1).
        LD      3,(2)           ;LOAD AC3 WITH F(M).
        ST      3,3(2)          ;REPLACE F(M-1) WITH F(M).
        LD      3,1(2)          ;LOAD AC3 WITH G(M-1).
        ST      3,2(2)          ;REPLACE G(M-2) WITH G(M-1).
        ST      1,1(2)          ;REPLACE G(M-1) WITH G(M).
        RXCH    Ø,1             ;PUT G(M) INTO ACØ.
        RTS     Ø               ;RETURN.
```

```
;       CONSTANT VALUES FOR THE FILTER ("FILT") SUBROUTINE ARE LISTED
;       BELOW. THE LABELS FOR THESE CONSTANTS BEGIN WITH A "L" TO
;       INDICATE THAT THEY ARE DECIMAL VALUES. THE CONSTANTS "A", "B"
;       AND "C" ARE THE FILTER COEFFICIENTS.
;
D10000: .WORD   10000
D5000:  .WORD   5000
A:      .WORD   675
B:      .WORD   11430
C:      .WORD   4128
;
;
;       CHANNEL 4 RAINFALL RATE ROUTINE FOLLOWS. RAIN ACCUMULATOR
;       READING IS TAKEN ONCE EVERY 4-MINUTES. SINCE .01 INCH OF
;       PRECIPITATION IS MINIMUM SENSITIVITY OF ACUMULATOR, THE MINIMUM
;       DETECTABLE RAIN RATE IN INCHES/HOUR IS (60/4) X .01 = .15 IN/HR.
;       SAMPLING OF ACCUMULATOR IS CONTROLLED BY A BIT (BIT 14) WHICH IS
;       DERIVED FROM A 1-PULSE/4-MIN. SIGNAL.

RAIN:   LD      0,4             ;FETCH RAIN ACCUMULATOR OUTPUT.
        ST      0,160
        SHR     0,4             ;TRUNCATE TO 8 BITS.
        CLRBIT  10              ;REMOVE CONTROL BIT.
        SETBIT  11              ;DATA POSITIVE.
        ST      0,22            ;STORE INSTANTANEOUS ACCUMULATOR
;                                OUTPUT.
        CLRBIT  11
        JSR     BINBCD
        LI      1,6             ;SET AC1 TO ADDRESS ACCUM DISPLAY.
        JSR     OUTPUT
        LD      0,160
        SKBIT   14              ;SKIP IF CONTROL BIT IS HIGH.
        JMP     GRAD            ;GO TO POTENTIAL GRADIENT ROUTINE.
        PFLG    3               ;RESET CONTROL F-F.
        CLRBIT  14              ;CLEAR CONTROL BIT FROM AC0.
        ST      0,4             ;REWRITE MEM AFTER CONTROL BIT REMOVED.
        SHR     0,4
        RCPY    0,2
        SUB     0,161           ;SUBTRACT PREVIOUS ACCUM READING FROM
;                                PRESENT READING.
        RCPY    0,1
        BOC     2,.+4           ;SKIP NEXT 3 INST IF AC0>0.
        ST      0,161           ;IF AC0<0, NEXT 3 INST WILL CHANGE
        LD      1,D256          ;CONTENTS TO AN EQUIVALENT POSITIVE
        ADD     1,161           ;VALUE.
        MPY     D15             ;MULTIPLY ACCUM DIFFERENCE BY (60/4)=15
;                                TO ADJUST TO HOURLY RATE.
        RCPY    1,0
        SETBIT  11              ;DATA POSITIVE.
        ST      0,20            ;STORE HOURLY RATE.
        CLRBIT  11
        ST      2,161           ;STORE PRESENT ACCUM READING.
        JSR     BINBCD
        LI      1,4             ;SET AC1 TO ADDRESS RAIN DISPLAY.
        JSR     OUTPUT
```

```
;       CHANNEL 5 POTENTIAL GRADIENT ROUTINE FOLLOWS.  DIGITAL INPUT
;       TO MICROPROCESSOR COVERS 0-999 AS GRADIENT RANGES BETWEEN -10000
;       VOLTS/METER TO 10000 V/M.  SOFTWARE ADDS BIAS AND SCALING TO
;       ADJUST INPUT SO OUTPUT WILL REFLECT ACTUAL GRADIENT.
;
GRAD:   LD      0,5             ;FETCH POTENTIAL GRADIENT DATA.
        SHR     0,2             ;TRUNCATE TO 10 BITS.
        SHL     0,1             ;MULTIPLY DATA BY 2 (SCALE).
        SKG     0,D999          ;SKIP IF GRADIENT POSITIVE.
        JMP     NEG
        SUB     0,D999          ;BIAS DATA.
        SKG     0,D999          ;SKIP NEXT INST IF GRAD>9990 V/M.
        JMP     SET
        LD      0,D999          ;SET GRAD=9990 V/M. THIS ELIMINATES
;                                PROBLEMS CAUSED WHEN BIT FLICKER MAKES
;                                GRADIENT > 9990 V/M.
SET:    SETBIT  11               ;DATA POSITIVE.
        JMP     STOR
NEG:    ST      0,170
        LD      0,D999
        SUB     0,170           ;ADJUST DATA IF GRADIENT NEGATIVE.
STOR:   ST      0,21            ;STORE ADJUSTED DATA.
        CLRBIT  11
        JSR     @BNBCD1
        LI      1,5             ;SET AC1 TO ADDRESS GRADIENT DISPLAY.
        JSR     OUTPUT
;
;
;       LAMP TEST ROUTINE FOLLOWS.  THIS ROUTINE OUTPUTS A 4-BIT TEST
;       CODE (1010) TO EACH DISPLAY DIGIT WHICH CAUSES EACH LED ELEMENT
;       TO LIGHT. IT ALSO SERVES AS A DIAGNOSTIC SINCE ALL HARDWARE BE-
;       TWEEN THE MICROPROCESSOR AND THE DISPLAYS IS EXERCISED. THE ROU-
;       TINE IS INITIATED BY DEPRESSING A FRONT PANEL SWITCH WHICH ACTI-
;       VATES A JUMP CONDITION FLAG, AND IT WILL CYCLE AS LONG AS THE
;       SWITCH REMAINS CLOSED.  RELEASING THE SWITCH RETURNS THE MICRO-
;       PROCESSOR TO THE DATA MODE.
;
LPTST:  LI      0,16            ;INITIALIZE REGISTERS.
        ST      0,180
        LI      1,0
HERE:   BOC     14,.+2          ;EXERCISE LAMP TEST IF FLAG IS ACTIVE,
        JMP     @STRT           ;OTHERWISE RETURN TO START.
        LD      0,HAAA          ;PUT TEST CODE (HEX 0AAA) IN AC0.
        RADD    1,0             ;PUT DISPLAY ADDRESS IN AC0.
        LI      3,0
        ROUT    8               ;OUTPUT CODE TO SELECTED DISPLAY.
        DSZ     180             ;SUBTRACT 1 FROM NUMBER IN LOC 180 AND
;                                SKIP NEXT INST IF NUMBER=0.
        JMP     HERE1
        JMP     LPTST
HERE1:  ADD     1,H1000         ;INCREMENT DISPLAY ADDRESS.
        JMP     HERE            ;CONTINUE.
;
;
;       CONSTANT VALUES USED IN RAINFALL RATE, POTENTIAL GRADIENT,
;       AND LAMP TEST ROUTINES ARE LISTED BELOW. THOSE WHOSE LABEL
;       BEGINS WITH A "D" ARE DECIMAL WHILE THOSE THAT BEGIN WITH
;       AN "H" ARE HEXADECIMAL.
```

```
D15:      .WORD    15
D256:     .WORD    256
D999:     .WORD    999
BNBCD1:   .WORD    61584
STRT:     .WORD    61440
H1000:    .WORD    0100
HAAA:     .WORD    0AAA
          .END
```